

(7)乱数の作成(rand関数)

プログラム

◆rand関数による乱数の作成方法を紹介します。

- ・乱数列の周期:2**19937-1
- ・ranuni関数に比べ周期が長く、生成された乱数列の性質が良い。

```
/*  
RAND関数を使用する場合は、分布を表すキーワードを第1引数として指定します。  
分布によってはそれぞれの分布に対応したパラメータ指定が必須となり、それらを第2引数以降に記述します。  
シード値の設定には専用のCALLルーチン CALL STREAMINITを使用します。  
*/
```

```
* ①;  
* rand関数では、分布を表すキーワードを第1引数として指定する;  
data work.rand1(keep=v1:);  
  call streaminit(100);  
  do I=1 to 10;  
    v1_1 = rand('uniform'); /* 一様乱数 */  
    v1_2 = rand('binomial',0.5,20); /* 2項分布に従う乱数。N=20,p=0.5 */  
    v1_3 = rand('normal',8,2); /* 平均が8、標準偏差が2の正規乱数 */  
    output;  
  end;  
run;
```

```
proc print data = work.rand1;  
run;
```

```
* ②;  
*「シード列は1種類」;  
*v2 とv3 は異なる乱数列が生成される;  
data work.rand2(keep=v2: v3:);  
  call streaminit(100);  
  do I=1 to 5;  
    v2_1 = rand('uniform'); /* 一様乱数 */  
    v2_2 = rand('binomial',0.5,20); /* 2項分布に従う乱数。N=20,p=0.5 */  
    v2_3 = rand('normal',8,2); /* 平均が8、標準偏差が2の正規乱数 */  
  
    v3_1 = rand('uniform'); /* 一様乱数 */  
    v3_2 = rand('binomial',0.5,20); /* 2項分布に従う乱数。N=20,p=0.5 */  
    v3_3 = rand('normal',8,2); /* 平均が8、標準偏差が2の正規乱数 */  
    output;  
  end;  
run;
```

```
proc print data = work.rand2;  
run;
```

結果

①で求めた乱数v1_1~v1_3と②で求めた乱数v2_1~v2_3、v3_1~v3_3を比較します。
シード列は1種類であるため、v1の値が、v2、v3に交互に入っていることが分かります。

①出力結果

OBS	v1_1	v1_2	v1_3
1	0.92848	11	5.4193
2	0.10486	8	8.4805
3	0.21541	9	10.4408
4	0.77911	5	5.0809
5	0.72309	14	9.6954
6	0.67381	12	5.6911
7	0.08477	10	9.8169
8	0.05068	10	9.9154
9	0.57531	11	7.9555
10	0.62104	9	6.7914

②出力結果

OBS	v2_1	v2_2	v2_3	v3_1	v3_2	v3_3
-----	------	------	------	------	------	------

1	0.92848	11	5.4193	0.10486	8	8.48046
2	0.21541	9	10.4408	0.77911	5	5.08091
3	0.72309	14	9.6954	0.67381	12	5.69107
4	0.08477	10	9.8169	0.05068	10	9.91539
5	0.57531	11	7.9555	0.62104	9	6.79138

※SASは、米国及び各国での米国SAS Institute Inc.の登録商標です。